Holy City Audio Forum

Former Home of SpinCAD Designer
Skip to content

Search... Search
Advanced search

[Moderator Control Panel]

External control of attack and release for Envelope Block

Post a reply

Search this topic... Search

7 posts • Page **1** of **1**

- Edit post (./posting.php?mode=edit&f=42&p=2594)
- Delete post (./posting.php?mode=delete&f=42&p=2594)
- Report this post (./report.php?f=42&p=2594)
- Information (./mcp.php?i=main&mode=post_details&f=42&p=2594&sid=d6d1dcc4501596550704606791e62b4c)
- Reply with quote (./posting.php?mode=quote&f=42&p=2594)

External control of attack and release for Envelope Block (#p2594)

by **Digital Larry** » Wed Dec 23, 2015 10:18 pm

OK the magical thing that needed to happen to get this started was adding frequency control to the 1 pole RDFX LPF, as follows:

```
@name 'LPF 1P'
@color "0x24f26f"
@audioInput input Input
@controlInput freqControl Frequency
@audioOutput lpf1 Output

equ freq 0.15
// variable - Name - low - high - multiplier - precision - option
@sliderLabel freq 'Frequency (Hz)' 80 2500 1000.0 1000.0 1 LOGFREQ

equ output Reg0
equ lpf1 Reg1

@isPinConnected Input

rdax input, 1.0
@isPinConnected Frequency
rdax lpf1, -1.0
mulx freqControl
rdax lpf1, 1.0
@else
rdfx lpf1, freq
@endif
wrax lpf1, 0

@setOutputPin Output lpf1
@endif
```

As noted before, with frequency control you can't use RDFX any more.

The algorithm implements the low pass filter with a single "K2" value shown here:
http://www.spinsemi.com/knowledge_base/ ... le_filters (http://www.spinsemi.com/knowledge_base/effects.html#Simple_filters)

In order to incorporate this into the envelope follower, I think I'd like to have max/min rise/fall times on the attack/decay filters settable by sliders. So that way you just connect up your 0.0 to 1.0 pot and you get the range you want.

Next I need to figure out how to map the control input value to the rise time. This will be a new slider value display function to incorporate into SpinCAD Builder. Haven't been spending much time with SpinCAD lately, need to get my mind wrapped around it again.

Again from the knowledge base notes,

For a given sample period t and a desired -3dB cutoff frequency of F, the value for K1 is equal to e^-(2πFt).

$K1 = e^{-(2\pi Ft)}$
$1 - K2 = e^{-(2\pi Ft)}$
$\ln(1 - K2) = -(2\pi Ft)$

alternative forms:
K2 = 1 - e^-(2πFt)
F = ln(1 - K2)/(-2πt)

Where K2 is the control input value and t is the sampling period.

The envelope follower filters go down to 0.51 Hz currently, which I think is the lower limit based on the smallest possible non zero coefficient for RDFX.

So, suppose that fs = 32768, then t = 30.518 microseconds.
At F = 0.5 Hz, then 1 - e^-(2πFt) = 0.000095869
At F = 1.0 Hz, then 1 - e^-(2πFt) = 0.0001917292
At F = 10 Hz, then 1 - e^-(2πFt) = 0.0019156384

etc.

Let's be careful here though! The multiplier parameter for RDFX is S1.14, so the smallest non zero coefficient is going to be 2^-14. Whereas the pot control is only 11 bits, so if the pot control went directly to the filter, the lowest possible non zero value would be 2^-11. So we should definitely scale down the control input with another multiplication prior to applying it to the filter.

Here's the controlled filter's Spin ASM:

```
rdax input, 1.0
rdax lpf1, -1.0
mulx pot0
rdax lpf1, 1.0
rdfx lpf1, freq
wrax lpf1, 0
```

Since the goal is to scale down the value following the mulx instruction, and this has (input - lpf1) going into it, then I would argue that instead of multiplying the pot by a fixed value, storing that and then retrieving it for the mulx, that we can apply the fixed value instead to the coefficients for the first two rdax instructions instead. So, the "fastest" setting of the filter (from the control panel slider) would then set that factor, and the pot would scale down from there. I think I will give that a try and see what happens.

For the rise time calculation, using this as a reference:

https://en.wikipedia.org/wiki/Rise_time (https://en.wikipedia.org/wiki/Rise_time)

tr =~ (approx) 0.34/Fc.

So,

Fc tr
0.5 Hz 0.68 s
1.0 Hz 0.34 s
10 Hz 0.034 s
25 Hz 0.0136 s

Top
- Edit post (./posting.php?mode=edit&f=42&p=2595)
- Delete post (./posting.php?mode=delete&f=42&p=2595)
- Report this post (./report.php?f=42&p=2595)
- Information (./mcp.php?i=main&mode=post_details&f=42&p=2595&sid=d6d1dcc4501596550704606791e62b4c)
- Reply with quote (./posting.php?mode=quote&f=42&p=2595)

Re: External control of attack and release for Envelope Bloc (#p2595)

by **Digital Larry** » Thu Dec 24, 2015 11:27 am

I was talking before about the structure of the follower for separate attack and decay time constants.

With low pass filters in series, and the decay generally longer than the attack, it's probably OK. I think this is the current structure. If there is a long attack you can't get a fast decay. I have covered this up somewhat by making the decay range longer (lower frequency) than the attack time.

With the low pass filters in parallel, and a maxx instruction to pick off the higher value of the two, then the attack time would not affect the decay time. However, the follower would not switch to the decay envelope until the attack envelope had come down to where the decay filter was going. I think I can give the decay filter "instant attack" by forcing the attack filter output value into the decay filter's memory register when the attack value is greater than the stored decay value.

As far as slow attack/fast decay envelopes, I don't think that a simple filter arrangement can do it. Each filter stage has the same attack and decay response. I don't mean as each other, but what I mean is that the attack filter's time constant in and of itself is the same whether it is going up or down. So, if the attack filter goes slower than the decay filter, the envelope follower would need to track max(attack, decay) going up and min(attack,decay) going down. There will need to be some trigger or level decision made per sample cycle to see which way we are going. I'm generally concerned about possible glitching or weird transitions in such a system but won't know until I try.

Top
- Edit post (./posting.php?mode=edit&f=42&p=2596)
- Delete post (./posting.php?mode=delete&f=42&p=2596)
- Report this post (./report.php?f=42&p=2596)
- Information (./mcp.php?i=main&mode=post_details&f=42&p=2596&sid=d6d1dcc4501596550704606791e62b4c)
- Reply with quote (./posting.php?mode=quote&f=42&p=2596)

Re: External control of attack and release for Envelope Bloc (#p2596)

by **Digital Larry** » Fri Dec 25, 2015 12:31 pm

F = ln(1 - K2)/(-2πt)

Where K2 is the control input value and t is the sampling period.

also:
F = Fs * (ln(1 - K2)/(-2π))

If we want to limit the range where this is applicable, let's use π/4 which is 1/4 of the sampling frequency, just above 8 kHz for 32.768 kHz sampling rate.

ln(1 - K2)/-(2π) = 1/4

ln(1 -K2) = -π/2
1 - K2 = e^(-π/2)
K2 = 1 - e^(-π/2)
K2 =~ 0.792121187

When K2 = 0, the expression evaluates to Fs * ln(1)/(-2π), but ln(1) = 0,0 so F = 0. So practically speaking we do not want K2 to go all the way to zero. K2 should go between something like

Top
- Edit post (./posting.php?mode=edit&f=42&p=2597)
- Delete post (./posting.php?mode=delete&f=42&p=2597)
- Report this post (./report.php?f=42&p=2597)
- Information (./mcp.php?i=main&mode=post_details&f=42&p=2597&sid=d6d1dcc4501596550704606791e62b4c)
- Reply with quote (./posting.php?mode=quote&f=42&p=2597)

Re: External control of attack and release for Envelope Bloc (#p2597)

by **Digital Larry** » Fri Dec 25, 2015 9:52 pm

I'm going to look really closely at the code in Spin's envelope filter patch.

```
;Now do wah, a 2 pole LPF, peaking.
;begin by getting control level into detector:

rdax   mono,1      ;get input
absa         ;absolute value
rdfx   avg,0.01      ;average input level
wrax   avg,0      ;write avg level, pass on
rdax   lavg,0.001
sof   -0.01,0
rdax   lavg,1
wrax   temp,0
rdax   avg,1
maxx   temp,1      ;filter a long average
wrax   lavg,0

;now set up a means by which the sensitivity control can affect the filter frequency:

rdax   lavg,1
sof    1,0.002      ;never let lavg go to zero
log   1,0
wrax   temp,0
rdax   avg,1
log   1,0
rdax   temp,-1
mulx    pot1
exp   1,0
rdfx   ffil,0.0005
wrax   ffil,1
sof   0.7,0.02      ;limit frequency range
```

```
wrax   wf,0
```

We'll take this bit by bit. And I only think this is what is going on.

```
rdax   mono,1      ;get input
absa         ;absolute value
rdfx   avg,0.01      ; average input level with low pass filter.  k2 = 0.01, Fo/Fs = ln(0.99)/(-2* pi)
wrax   avg,0      ;write avg level, pass on
```

Take the input, rectify it.
average input level with low pass filter.
k2 = 0.01, Fo/Fs = ln(0.99)/(-2* pi)
at 32 kHz implied, Fo = 52.4 Hz.

This is the "attack" filter.

Next we find the decay part.

```
rdax   lavg,0.001
sof   -0.01,0
rdax   lavg,1
wrax   temp,0
rdax   avg,1
maxx   temp,1      ;filter a long average
wrax   lavg,0
```

We see this variable "lavg", what is it? Don't know yet.

We read it with a multiplier of 0.001, scale it again by -0.01 and then read it again adding in at a gain of 1.0. I'm not quite sure why he didn't use:

```
rdax lavg, 0.99999
wrax temp, 0              ; save slowly decaying "lavg" to temp
```

Next, (this is clever), write to "lavg" with the larger of the attack filter output or the "lavg" signal from last time multiplies by 0.99999. So this is the "instant attack" behavior for the decay section.

```
rdax   avg,1
maxx   temp,1      ; temp contains lavg * 0.99999
wrax   lavg,0
```

Here, the coefficient is "k1" :

k1 = e^-(2πF/Fs) = 0.99999
F0/Fs=ln(k1)/(-2 * pi)
at 32 kHz, Fo = 0.52 Hz

Top
- Edit post (./posting.php?mode=edit&f=42&p=2598)
- Delete post (./posting.php?mode=delete&f=42&p=2598)
- Report this post (./report.php?f=42&p=2598)
- Information (./mcp.php?i=main&mode=post_details&f=42&p=2598&sid=d6d1dcc4501596550704606791e62b4c)
- Reply with quote (./posting.php?mode=quote&f=42&p=2598)

Re: External control of attack and release for Envelope Bloc (#p2598)
by **Digital Larry** » Fri Dec 25, 2015 10:32 pm

Next,

```
rdax   lavg,1
sof    1,0.002      ;never let lavg go to zero
log   1,0
wrax   temp,0
```

Take lavg (the attack/decay envelope from the input signal).
Add 0.002 to it.
Take the log of this value. Limited by 0.002 factor.
Write result to temp.

What's going on here? Don't have any idea.

Next,

```
rdax   avg,1
log   1,0
rdax   temp,-1
mulx    pot1
exp   1,0
```

We take the value of the attack filter.
We take its log. No offset as with decay output.
We subtract temp, which is the (log of the decay + 0.002)
We multiply that by the pot1 value (potx in general)
Calculate the exponent of the result.

So, I remember that:

exp(log(a) - log(b)) = a/b

So if you take the mulx instruction out for the time being, we're dividing the attack signal by the attack+decay signal. When the attack signal is rising, these are both the same (other than the 0.002 offset). So, on the note attack, the result is a/(a+0.002). When a = 0.002, the result is 0.5. As a goes towards 1.0, the ratio a/(a + 0.002) goes very closely to 1.0.

When the attack signal decays, then it goes below the attach+decay signal. So let's suppose that attack has gone to 0.0001 and decay is still up there at 0.7 or so, 0.0001/0.7 = 0.000143. So, as far as I can tell, this is mostly a peak detection circuit that pulls the peak out of a wide range of possible input levels.

To be honest, I can see why the division is done on the attack portion - so that a detected attack, above a certain threshold (0.002 in this case) gives a strong attack pulse over a wide range of input signals. What's not so clear to me is whether this divided signal has much relevance during the decay portion.

Next, including the mulx instruction before the exp instruction...

Recall that 10^(1/2) = square root of 10. So, multiplying a number by a pot value (always less than one) prior to the exp instruction amounts to a "variable root" function. As the pot value decreases, say to 0.1, that represents the 10th root of the peak detected signal. Again, these values are all between 0.0 and 1.0, so a higher root tends to compress the signal upwards towards 1.0. So, as the pot value goes down, the dynamic range of the signal decreases, but the maximum (if it went all the way to 1.0) stays the same.

Top
- Edit post (./posting.php?mode=edit&f=42&p=2599)
- Delete post (./posting.php?mode=delete&f=42&p=2599)
- Report this post (./report.php?f=42&p=2599)
- Information (./mcp.php?i=main&mode=post_details&f=42&p=2599&sid=d6d1dcc4501596550704606791e62b4c)
- Reply with quote (./posting.php?mode=quote&f=42&p=2599)

Re: External control of attack and release for Envelope Bloc (#p2599)

by **Digital Larry** » Sat Dec 26, 2015 7:56 am

Phew. We're almost done. And I hope I'm at least close to being correct.

```
rdfx   ffil,0.0005
wrax   ffil,1
sof   0.7,0.02      ;limit frequency range
wrax   wf,0
```

The first two instructions are yet another low pass filter applied to the processed envelope.

Here, K2 = 0.0005, so Fo = Fs * (ln(1 - K2)/(-2π)) = 2.61 Hz for 32768 Hz sampling rate.

The SOF instruction adjusts the final processed envelope signal to a desired control range for the filter, which is undoubtedly best done by ear,

Top
- Edit post (./posting.php?mode=edit&f=42&p=2600)
- Delete post (./posting.php?mode=delete&f=42&p=2600)
- Report this post (./report.php?f=42&p=2600)
- Information (./mcp.php?i=main&mode=post_details&f=42&p=2600&sid=d6d1dcc4501596550704606791e62b4c)
- Reply with quote (./posting.php?mode=quote&f=42&p=2600)

Re: External control of attack and release for Envelope Bloc (#p2600)

by **Digital Larry** » Sat Dec 26, 2015 1:14 pm

I just did a little simulation comparison between the 2 envelope followers with the 2-pole resonant low pass filter.

The basic SpinCAD envelope follower, which is just two low pass filters with a maxx instruction between them, can sound ok but starts to sound bad when the attack time gets too fast, above around 7 Hz. I believe this is due to the fact that the attack filter level is usually above the decay filter level, even when the signal is decaying, so you get some sub-audio filter modulation. I think I will add the peak follower strategy of the Spin enveloper follower to prevent this. The envelope level starts low and goes high, decaying to low again.

Spin's envelope follower on the other hand, starts high. If you turn the sensitivity to zero then the filter is wide open. The envelope dips down in response to peaks then goes back up as the signal decays.

These are two fundamentally different envelope sounds, so I'll keep both blocks - Envelope and Envelope II or something. I tried a quick patch with one envelope-filter combination going left, and the other going right. Sounded nice.

Now I have to figure out where to allow adjustments.

Top

Display posts from previous: All posts Sort by Post time Ascending Go

Post a reply

7 posts • Page **1** of **1**

Return to Developer's Corner

Jump to: Developer's Corner Go

Quick-mod tools: Lock topic Go

Who is online

Users browsing this forum: Digital Larry and 0 guests

Administration Control Panel