Holy City Audio Forum

Former Home of SpinCAD Designer

Skip to content

Search... Search

Advanced search

[Moderator Control Panel]

Reverb block!

Post a reply

Search this topic... Search

13 posts • Page **1** of **1**

- Edit post (./posting.php?mode=edit&f=31&p=1834)
- Delete post (./posting.php?mode=delete&f=31&p=1834)
- Report this post (./report.php?f=31&p=1834)
- Warn user (./mcp.php?i=warn&mode=warn_post&f=31&p=1834&sid=befaf904784df501ab581b011c284453)
- Information (./mcp.php?i=main&mode=post_details&f=31&p=1834&sid=befaf904784df501ab581b011c284453)
- Reply with quote (./posting.php?mode=quote&f=31&p=1834)

Reverb block! (#p1834)

by **disasterarea** » Sun Sep 28, 2014 4:40 pm

Larry, first of all thanks for all the work on the phaser block, it's awesome. Truly great. The filter and envelope blocks are coming along nicely, too. I think that the reverb block could use some tweaking, right now we just have the "simple reverb" block.

```
;  Program: Render Block exported from SpinCAD Designer
;------ Input
;------ Pot 0
;------ Minimum reverb
RDAX ADCL,0.2500000000
RDA 122,0.325
WRAP 0,-1.0
RDA 426,0.325
WRAP 123,-1.0
RDA 980,0.325
WRAP 427,-1.0
RDA 1903,0.325
WRAP 981,-1.0
WRAX REG0,1.0000000000
RDA 21990,1.0
MULX POT0
RDA 5727,-0.325
WRAP 1904,0.325
WRA 5728,1.99
RDAX REG0,1.0000000000
RDA 12240,1.0
MULX POT0
```

```
RDA 16973,-0.325
WRAP 12241,0.325
WRA 16974,1.99
WRAX REG1,0.0000000000
;------ Output
RDAX REG1,1.0000000000
WRAX DACL,0.0000000000
```

So that's the simple reverb. The patch here is input -> simple reverb -> output. Pot 0 connects to the control input. Looking at the code, the pot input tweaks the reverb time and the mix. This is great for a minimum implementation but bad if we're trying to do a really nice reverb.

I always prefer to do my own mixing, it lets me get much better control. This is pie in the sky talking here, but I'm thinking the ability to specify the delay lengths and coefficients for each section, plus a control input for the "smearing" would be all we really need. All the reverbs are pretty much the same, only the coefficients and delay lengths are different.

Looking at the "rev_rt_d_f.spn," there's a variable called kdiff that is the diffusion constant, and there's a filter after the reverb. The filter isn't all that important since we can use the shelving or resonant filters to tailor the 'verb, but the diffusion is useful.

There's another nice reverb, the "dance_ir_h_l.spn" by Dave Spinkler. One of the super-cool tricks he uses is creating a reverb "freeze" by increasing the reverb time and decreasing the input gain. This makes the reverb go on forever but no new sounds get reverb'd.

```
;prepare pots to affect control variables:
;pot0 controls reverb time, but also affects input drive level;
;reveb time is moderate up to about mid position, then increases
;to infinity (or nearly) at full position.
;input drive is constant, but decreases at the full pot0 position.
;output mix is varied over the first half of pot0, then remains
;high to the end of pot0's range.

rdax   pot0,1.999   ;get pot0, clip the upper half of pot0's range.
wrax   kmix,0      ;write the output mix value

rdax   pot0,-1      ;get pot0 again, 0 to -1
sof   1,0.999      ;now +1 to 0
sof   1.999,0      ;now +1 until midpint, then decreases to 0
wrax   kin,0      ;write the input attenuator value

rdax   pot0,1      ;get pot0 again
wrax   krt,1      ;save in krt, keep in ACC
sof   1,-0.5      ;subtract 1/2
skp   gez,2      ;skp if pot is in upper half of range
sof   0,0.5      ;load accumulator with +0.5
wrax   krt,0      ;overwrite if pot is in lower half of range
```

How cool is that? Super cool. One pot handles the output mix, input gain, and reverb time. Master level.

Anyway, the cool thing is we can mimic a lot of that with SOF operations, really we just need the ability to make the reverb time 1.0 or really close to it.

That's a lot of stuff to take in, I know. But I think being able to do a good reverb is super-important - that was pretty much the whole reason behind the FV-1's architecture. Thoughts?

Top

- Edit post (./posting.php?mode=edit&f=31&p=1835)
- Delete post (./posting.php?mode=delete&f=31&p=1835)
- Report this post (./report.php?f=31&p=1835)
- Information (./mcp.php?i=main&mode=post_details&f=31&p=1835&sid=befaf904784df501ab581b011c284453)
- Reply with quote (./posting.php?mode=quote&f=31&p=1835)

Re: Reverb block! (#p1835)

by **Digital Larry** » Sun Sep 28, 2014 9:07 pm

Hello,

Thanks for the kind words. I haven't focused much on the reverb algorithms up to now, obviously. What it takes is going through each algorithm that seems to have "magic" numbers in it, and then:
1) Determine what the usable range of that magic number is
2) Figure out what sort of "curve" or "taper" it should have on the control panel slider, i.e. so that you don't wind up with no action for most of a slider's travel, then all the excitement is way over on one end of it. For example, I try to have filter frequencies represented on a logarithmic scale.
3) Figure out whether it wants to have a control input, or whether just having a control panel slider is good enough.

Now you could wait for me to figure it out, or if you wanted to help, you might try some of these algorithms and then map out some of the answers to these questions. That could jump start the process.

Top

- Edit post (./posting.php?mode=edit&f=31&p=1836)
- Delete post (./posting.php?mode=delete&f=31&p=1836)
- Report this post (./report.php?f=31&p=1836)
- Warn user (./mcp.php?i=warn&mode=warn_post&f=31&p=1836&sid=befaf904784df501ab581b011c284453)
- Information (./mcp.php?i=main&mode=post_details&f=31&p=1836&sid=befaf904784df501ab581b011c284453)
- Reply with quote (./posting.php?mode=quote&f=31&p=1836)

Re: Reverb block! (#p1836)

by **disasterarea** » Mon Sep 29, 2014 4:34 am

I'd love to help!

Here's the OEM program 2, the "small hall." The stock program has controls for predelay (pot1,) reverb time (rt, pot0,) and damping (pot2.) Pre-delay happens before the reverb so we can safely just put in our own delay block for that, no need to mess with it. Just be sure to put smoothing on the delay control so it doesn't get all squirrely when you spin the pot!

I'm thinking control panel sliders for:
AP ring sizes (iap and ap)
Reverb delay blocks
HPF and LPF filter coefficients

Control inputs for:
Reverb Time (rt)
Damping crossfade (shown as pot2 here, kdiff in other algorithms)
LFO injection for all-pass smearing (skip the final 12 lines if not connected.)

No need to worry about "magic" numbers, just let us tweak the sliders ourselves and that'll be more than good enough. the algorithm below is pretty much the same as all the other Spin 4-delay + AP ring reverbs, only the coefficients are changed. Some of these reverbs can use a lot of delay RAM, if you're doing other delay tasks in the same algorithm you have to be careful how big you make the delay blocks.

```
mem   pdel   4600  ; pre-delay, 100mS at a 46khz clock freq.
mem   iap1   156  ; all-pass filters
mem   iap2   223
mem   iap3   332
mem   iap4   448
;
mem   ap1   1251  ; more all-passes
mem   ap1b   1751
mem   ap2   1443
mem   ap2b   1343
mem   ap3   1582
mem   ap3b   1981
mem   ap4   1274
mem   ap4b   1382
;
mem   del1   3559  ; actual reverb delay blocks
mem   del2   2945
mem   del3   3976
mem   del4   4445

equ    temp   reg0  ; locations for filtering
equ   hpf1   reg1
equ   hpf2   reg2
equ   hpf3   reg3
equ   hpf4   reg4
equ   lpf1   reg5
equ   lpf2   reg6
equ   lpf3   reg7
equ   lpf4   reg8
```

```
equ   rt   reg9  ; reverb time  --- this needs a control input
equ   iapout   reg10 ; holding register for all pass output
equ   pdelo   reg11 ; holding register for output of pre-delay

;constants:  ---- these should have control panel sliders

equ   kfh   0.01  ; high pass filter
equ   kfl   0.4  ; low pass filter
equ   kiap   0.5  ; all pass filter
equ   klap   0.6  ;  all pass filter

;prepare decay time pot:

rdax   pot0,1
sof   0.55,0.3
wrax   rt,0         ;rt ranges 0.3 to 0.85

;do variable predelay:

skp   run,1
wldr   rmp0,0,4096      ;initialize predelay

rdax   adcl,0.5      ;put inputs into predelay
rdax   adcr,0.5
wra   pdel,0

cho   rda,rmp0,reg|compc,pdel   ;get outputs from predelay, interpolated
cho   rda,rmp0,0,pdel+1
wrax   pdelo,0         ;write predelay output to register

cho   rdal,rmp0      ;read current predelay pointer
rdax   pot1,-0.5      ;subtract pot for servo control of pointer
wrax   rmp0_rate,0      ;maintain predelay pointer

;now run predelayed signal into 4 all passes:

rdax   pdelo,0.25      ;attenuate signal to avoid clipping
rda   iap1#,kiap
wrap   iap1,-kiap
rda   iap2#,kiap
wrap   iap2,-kiap
rda   iap3#,kiap
wrap   iap3,-kiap
rda   iap4#,kiap
wrap   iap4,-kiap
wrax   iapout,0      ;write to register for ring injection
```

```
;now do reverb ring, use temp as reg for filtering:

;aps into delay1:

rda   del4#,1      ;read previous delay
mulx   rt      ;multiply by reverb time coefficient
rdax   iapout,1   ;read left input from input allpass filter bank
rda   ap1#,klap   ;do an allpass filter
wrap   ap1,-klap
rda   ap1b#,klap   ;do second all pass filter
wrap   ap1b,-klap
wrax   temp,1      ;write to temp, keep in acc
rdfx   lpf1,kfl   ;low pass filter
wrlx   lpf1,-1
rdfx   hpf1,kfh   ;high pass filter
wrhx   hpf1,-1
rdax   temp,-1
mulx   pot2      ;crossfade between filter and no filter
rdax   temp,1
wra   del1,0      ;write to next delay, clear accumulator

;aps into delay2:

rda   del1#,1
mulx   rt
rdax   iapout,1
rda   ap2#,klap
wrap   ap2,-klap
rda   ap2b#,klap
wrap   ap2b,-klap
wrax   temp,1
rdfx   lpf2,kfl
wrlx   lpf2,-1
rdfx   hpf2,kfh
wrhx   hpf2,-1
rdax   temp,-1
mulx   pot2
rdax   temp,1
wra   del2,0

;aps into delay3:

rda   del2#,1
mulx   rt
rdax   iapout,1
rda   ap3#,klap
wrap   ap3,-klap
```

```
rda   ap3b#,klap
wrap   ap3b,-klap
wrax   temp,1
rdfx   lpf3,kfl
wrlx   lpf3,-1
rdfx   hpf3,kfh
wrhx   hpf3,-1
rdax   temp,-1
mulx   pot2
rdax   temp,1
wra   del3,0

;aps into delay4:

rda   del3#,1.0
mulx   rt
rdax   iapout,1
rda   ap4#,klap
wrap   ap4,-klap
rda   ap4b#,klap
wrap   ap4b,-klap
wrax   temp,1
rdfx   lpf4,kfl
wrlx   lpf4,-1
rdfx   hpf4,kfh
wrhx   hpf4,-1
rdax   temp,-1
mulx   pot2
rdax   temp,1
wra   del4,0

;take outputs as taps from reverb ring:

rda   del1,0.8
rda   del2+1876,1.5
rda   del3+2093,1.1
rda   del4+2793,1
wrax   dacl,0      ;write output, clear accumulator

rda   del1,0.8
rda   del2+923,1.5
rda   del3+1234,1.1
rda   del4+2267,1
wrax   dacr,0

;set up lfo, 1Hz to 2Hz, +/-100 samples, for smearing loop all passes:
```

```
skp   run,2
wlds   sin0,45,50
wlds   sin1,53,50

;smear 4 allpass filters in reverb ring:

; this is a chorus on the all pass stages, to simulate how echoes get
bounced around in the real world by air currents and to reduce the
repetitive "echo" sound.  Probably needs a checkbox in the control panel
and / or an input for an LFO.  Maybe if no LFO is connected, skip these
lines.

cho    rda,sin0,reg|compc,ap1+50   ;sin0
cho   rda,sin0,0,ap1+51
wra   ap1+100,0

cho    rda,sin0,cos|compc,ap2+50   ;cos0
cho   rda,sin0,cos,ap2+51
wra   ap2+100,0

cho    rda,sin1,reg|compc,ap3+50   ;sin1
cho   rda,sin1,0,ap3+51
wra   ap3+100,0

cho    rda,sin1,cos|compc,ap4+50   ;cos1
cho   rda,sin1,cos,ap4+51
wra   ap4+100,0
```

Top

- Edit post (./posting.php?mode=edit&f=31&p=1837)
- Delete post (./posting.php?mode=delete&f=31&p=1837)
- Report this post (./report.php?f=31&p=1837)
- Information (./mcp.php?i=main&mode=post_details&f=31&p=1837&sid=befaf904784df501ab581b011c284453)
- Reply with quote (./posting.php?mode=quote&f=31&p=1837)

Re: Reverb block! (#p1837)

by **Digital Larry** » Mon Sep 29, 2014 5:08 am

What I meant about "magic numbers" was simply the parameters under control. So for example, in the case of all pass length, it's unlikely that going all the way to zero is going to be useful. I'll try to goof around with this a bit today and see what happens (just changing things in the ASM directly).

What I'm curious to find out is whether there really is a wide range of useful adjustments, or whether the numbers we see in the supplied code examples represent the end result of lots of trial and error up front to find the optimum setting.

I'm also OK including pre-delay as part of the block, which can drop out any code generation if you set it to zero. The more I use the program myself for generating patches (which I've only really been doing seriously for a couple of months), and I find myself doing the same things over and over (like adding a feedback path in a delay) then I am tempted to add that capability to the block, while allowing that sometimes I won't want to use it, and I don't want to be penalized code-wise if that's the case.

Just as an example of that, right now I can only detect whether input pins have a connection. However, say in the case of the three tap delay, I'd like to drop code generation for delay taps whose OUTPUT is not connected. I think I have a way of doing that but haven't tried it yet.

Stay tuned and thanks for the idea!

DL

Top
- Edit post (./posting.php?mode=edit&f=31&p=1838)
- Delete post (./posting.php?mode=delete&f=31&p=1838)
- Report this post (./report.php?f=31&p=1838)
- Warn user (./mcp.php?i=warn&mode=warn_post&f=31&p=1838&sid=befaf904784df501ab581b011c284453)
- Information (./mcp.php?i=main&mode=post_details&f=31&p=1838&sid=befaf904784df501ab581b011c284453)
- Reply with quote (./posting.php?mode=quote&f=31&p=1838)

Re: Reverb block! (#p1838)

by **disasterarea** » Mon Sep 29, 2014 9:50 am

I honestly have no idea how the AP and delay tap sizes are chosen. There was some discussion over on the Spin forum about it, and the gist I got from Keith Barr at the time was that he just dicks around with the numbers until he got something he liked. That's a very unsatisfying answer but it is what it is. I don't think zeroes are practical on the all-passes either, nor for the delay times for the actual multi-tap section.

I guess we could look at all the Spin-provided reverb examples and figure out what the coefficients and delay lengths are to see whether there is some logic or pattern, but I think it might be better to just give the designer access to the values directly. It's 12 sets of parameters for the APs and the delay, then HF and LF coefficients and possibly the pre-delay and chorus / smearing sections. That's a lot of stuff to mess with, maybe have it come up with the values from the "small hall" by default? That way at least you'd get something usable without tweaking.

Top
- Edit post (./posting.php?mode=edit&f=31&p=1839)
- Delete post (./posting.php?mode=delete&f=31&p=1839)
- Report this post (./report.php?f=31&p=1839)
- Information (./mcp.php?i=main&mode=post_details&f=31&p=1839&sid=befaf904784df501ab581b011c284453)
- Reply with quote (./posting.php?mode=quote&f=31&p=1839)

Re: Reverb block! (#p1839)

by **Digital Larry** » Mon Sep 29, 2014 5:33 pm

OK I took a look at this most recent algorithm and put it into a block. It somewhat goes against the grain of rationale for being a block in SpinCAD Designer - simply in that it's too big to fit much other functionality, and (to me) there's not much difference between tweaking parameter values in SpinCAD or directly editing them in the ASM. On the other hand, if we can construct all of this so that the algorithm I'm talking about is the MAXIMUM, and there could be other options (down to the

"Minimum" patch) that could all be controlled through control panel options, then I'd be happier. For example, this one has a stereo output, and I could save a dozen instructions or so simply by taking that out. It will require a fair amount of study of the different reverbs that are already published.

Some SpinCAD Designer shortcomings that this reverb algorithm helped to highlight:
a) Ongoing issues with simulation of Ramp LFO
b) No indication when a block or algorithm uses a built-in LFO. This means you can put down several blocks which use the same built-in LFO without any warning and then wonder why things aren't working.

Top

- Edit post (./posting.php?mode=edit&f=31&p=1841)
- Delete post (./posting.php?mode=delete&f=31&p=1841)
- Report this post (./report.php?f=31&p=1841)
- Information (./mcp.php?i=main&mode=post_details&f=31&p=1841&sid=befaf904784df501ab581b011c284453)
- Reply with quote (./posting.php?mode=quote&f=31&p=1841)

Re: Reverb block! (#p1841)

by **Digital Larry** » Tue Sep 30, 2014 8:08 am

Some initial analysis of the last algorithm:

Changing kiap and klap affects the reverb's smoothness. Below 0.2 the reverb starts to sound a bit lumpy. Taking it up to 0.9 doesn't seem to change much from the default settings, but it probably does.

The reverb time can be taken up to 0.95, at which point it really lasts a long time!

Reducing the kfh high pass parameter to 0.1 moves the high pass frequency up high enough that you only get really high frequencies coming through.

I also changed the delay line lengths downward, which I guess reduced the reverb time, but didn't make the reverb sound bad.

Top

- Edit post (./posting.php?mode=edit&f=31&p=1842)
- Delete post (./posting.php?mode=delete&f=31&p=1842)
- Report this post (./report.php?f=31&p=1842)
- Information (./mcp.php?i=main&mode=post_details&f=31&p=1842&sid=befaf904784df501ab581b011c284453)
- Reply with quote (./posting.php?mode=quote&f=31&p=1842)

Re: Reverb block! (#p1842)

by **Digital Larry** » Tue Sep 30, 2014 9:47 am

Check out this section of the code:

```
;take outputs as taps from reverb ring:

rda   del1,0.8
rda   del2+1876,1.5
```

```
rda   del3+2093,1.1
rda   del4+2793,1
rdax adcl, 1.0    ; dry mix
wrax   dacl,0      ;write output, clear accumulator

rda   del1,0.8
rda   del2+923,1.5
rda   del3+1234,1.1
rda   del4+2267,1
wrax   dacr,0
```

This is the output mix of four different taps per channel. If you wish to change the delay line size, then some decision needs to be made about these hard coded offsets.

My first inclination for the control panel is to allow selection of a reverb structure that can go between the "minimum" implementation to a more fleshed out one. Maybe select # of delay lines and all pass stages... kinda like I did with the phase shifter? I've been playing through this one, with ridiculous amounts of reverb time, and it's pretty fun. I think it sounds great. But I don't think the smaller one sounds bad by guitar standards.

Top

- Edit post (./posting.php?mode=edit&f=31&p=1843)
- Delete post (./posting.php?mode=delete&f=31&p=1843)
- Report this post (./report.php?f=31&p=1843)
- Information (./mcp.php?i=main&mode=post_details&f=31&p=1843&sid=befaf904784df501ab581b011c284453)
- Reply with quote (./posting.php?mode=quote&f=31&p=1843)

Re: Reverb block! (#p1843)

by **Digital Larry** » Tue Sep 30, 2014 11:18 am

Did a few experiments.

Cut the delays out one by one until only one was left.

Then I cut out the input all pass blocks one by one until just one was left (going from shorter to longer).

The SPN files are attached. You can get a decent reverb with half as many blocks and all-pass stages. Down at one each it gets a bit rougher and for some reason increasing the pre-delay causes a wild oscillation. Be careful! But if you want ze smoozest sound, well then you must have 4 delays and 4 all passes.

So the first stab at this is just going to allow you to select 1 to 4 delay ring stages and 2 to 4 input all passes. That gives a wide swath of resource usage and pretty good sound overall for most cases (which excludes 20-second decay times).

[sorry, attachment was corrupted]

Top

- Edit post (./posting.php?mode=edit&f=31&p=1847)
- Delete post (./posting.php?mode=delete&f=31&p=1847)
- Report this post (./report.php?f=31&p=1847)
- Warn user (./mcp.php?i=warn&mode=warn_post&f=31&p=1847&sid=befaf904784df501ab581b011c284453)
- Information (./mcp.php?i=main&mode=post_details&f=31&p=1847&sid=befaf904784df501ab581b011c284453)
- Reply with quote (./posting.php?mode=quote&f=31&p=1847)

Re: Reverb block! (#p1847)

by **disasterarea** » Tue Sep 30, 2014 4:53 pm

That's an awesome start. Being able to select the number of stages is pretty important, I think, but so is being able to change the values of each stage. The "reverse" reverb is one example, where the coefficients increase instead of decreasing, so the reverb builds up rather than decays. Perhaps a dedicated "reverse" style setup where the taps and filters get bigger, and then a normal one where they get smaller" I don't know how hard it is to implement a bunch of control panel business inside the block, but it would be cool to be able to play with values in real time (or close to it) via the simulator.

I'll take some time to flash these examples tomorrow and see what happens.

Top

- Edit post (./posting.php?mode=edit&f=31&p=1848)
- Delete post (./posting.php?mode=delete&f=31&p=1848)
- Report this post (./report.php?f=31&p=1848)
- Information (./mcp.php?i=main&mode=post_details&f=31&p=1848&sid=befaf904784df501ab581b011c284453)
- Reply with quote (./posting.php?mode=quote&f=31&p=1848)

Re: Reverb block! (#p1848)

by **Digital Larry** » Tue Sep 30, 2014 5:10 pm

The real answer is:

I'm trying to make adding stuff to control panels easier. At present I can easily add a slider to control a double parameter in the algorithm, like the all-pass coefficient. Beyond that it is not quite as easy so I have to spend some time expanding the slider creation functionality.

Writing the control panel code brute force in Java is only slightly more fun than dental surgery.

Top

- Edit post (./posting.php?mode=edit&f=31&p=1855)
- Delete post (./posting.php?mode=delete&f=31&p=1855)
- Report this post (./report.php?f=31&p=1855)
- Warn user (./mcp.php?i=warn&mode=warn_post&f=31&p=1855&sid=befaf904784df501ab581b011c284453)
- Information (./mcp.php?i=main&mode=post_details&f=31&p=1855&sid=befaf904784df501ab581b011c284453)
- Reply with quote (./posting.php?mode=quote&f=31&p=1855)

Re: Reverb block! (#p1855)

by **disasterarea** » Sun Oct 05, 2014 5:15 pm

I've been playing with the new ReverbA block and it's good, really good. I'll write up some patches and see what I can do with this.

How do we figure out what the lengths of the delay lines are? You can still get a great reverb with two delays if each delay is pretty long, but I don't see a way to change the lengths or the coefficients. I know that control panel stuff is gnarly for you, sorry

Top

- Edit post (./posting.php?mode=edit&f=31&p=1856)
- Delete post (./posting.php?mode=delete&f=31&p=1856)
- Report this post (./report.php?f=31&p=1856)
- Information (./mcp.php?i=main&mode=post_details&f=31&p=1856&sid=befaf904784df501ab581b011c284453)
- Reply with quote (./posting.php?mode=quote&f=31&p=1856)

Re: Reverb block! (#p1856)

by **Digital Larry** » Mon Oct 06, 2014 4:28 am

The full structure of this algorithm contains 17 separate delay buffers. My suggestion, if you want to do fundamental reverb algorithm research, is to get comfortable doing it right in the original Spin ASM. The Spin ASM that SpinCAD Designer generates suppresses all "mem" declarations and simply calculates all memory references explicitly. My goal with creating blocks can be summarized as follows, with wiggle room to change my mind for any or no reason:

1) Develop annotations for Spin ASM allowing easy creation of blocks and control panels handling many typical adjustment scenarios ("SpinCAD Builder" language).
2) Where the desired structural or computational complexity prevents easy implementation of a general purpose solution, do it begrudgingly in Java.

What I'd really like to shoot for is a set of typical algorithms with names that people can relate to, such as "Room", "Plate", "Hall", "Trashcan", etc. with a reasonably small set of typical, understandable adjustments for the control panel. I have not yet investigated enough different algorithms to categorize things this way, but I can tell you this:

1 - the "All Pass" coefficients for the input section (4 all passes in parallel) and ring all passes set the diffusion or smoothness of the sound. There is some discussion of this at the Spin Knowledge base. Lower values of this are more "roomy" and higher values are more "hally" and supposedly higher than that gets "ringy". There's possibly some optimum setting in the middle which you'll have to adjust to your own liking.
2 - one different suggested structure is to use a multitap delay as the feeder instead of a pre-delay/all-pass block. haven't tried this. might make it a control panel selection or maybe a different block completely.
3 - there may be other ring structures that are useful but all I'm planning to do about it for the near term is to look at available algorithms to get an idea of what they are.
4 - The "Minimum" reverb block supplied by Spin has 4 parallel all-passes in the input, no pre-delay, and two delay-line-single-all-passes in the reverb ring with no filtering for HF damping.

So, I MIGHT come up with reverb algorithm sub blocks, but I tend to doubt even that, at least short-term. Sorry to disappoint you!

Top

Display posts from previous: All posts Sort by Post time Ascending Go

Post a reply
13 posts • Page **1** of **1**
Return to SpinCADBlocks

Jump to: SpinCADBlocks	Go
Quick-mod tools: Lock topic	Go

Who is online

Users browsing this forum: Digital Larry and 0 guests